UnseenGDK

GDK Core Tutorial V.01

By John Onyon a.k.a Unseen Machine

It is assumed throughout this tutorial/manual that the reader has at least a basic knowledge of programming in either QB64, VisualBasic, QuickBasic or QBasic.

This library ONLY works with QB64 http://www.qb64.net/

Introduction

UnseenGDK is a free for non-commercial use 2D game development kit built for and with QB64. It contains commands for spriting, collision detection, vector movement, particle and tile engines and much more.

Designed to make making games simple, fast and most of all fun. UnseenGDK was built after I found myself writing similar routines for each new program. I have written the same mouse routine more times than I care to remember, and I thought, enough is enough, and started copy/pasting Subs/Functions from one program to the next. Then Clippy taught me about libraries, and with a lot of work UnseenGDK was born. 

Although I am responsible for what is in UnseenGDK and am also responsible for the majority of the code, CodeGuy must be thanked for all his work on optimising the commands, as must Zom-B for the image rotation function. Also, for help with testing GDK and in making this manual/tutorial OlDosLover needs a mention too.
I hope you enjoy using UnseenGDK, and if you do use it to make a game, please send me a picture or copy of your project, and also any ideas you have for UnseenGDK. Happy coding folks.

Unseen Machine

Setting up UnseenGDK

In order for your program to have access to the commands in UnseenGDK you must include a reference to the library. This reference is normally added at the bottom of you program. 

To include the library add this code to the bottom of your program:

REM $INCLUDE:’UnseenGDK.bm’

Once you have included the library you need to perform one more step in order to  access its commands, initialising the data types that UnseenGDK uses. The core of the GDK data type’s are initialised with one command.

To initialise the GDK data types, add this code to the start of your program:

GDK_Start

Now you can use the majority of the commands in UnseenGDK. 

The next step is to set up a screen where your program will be displayed. You can do this with the prebuilt GDK commands or with any other method you like, but make sure you ALWAYS use 32 bit colour mode.  

Here is how you would normally create a 32 bit, 800 * 600 non full-screen mode program screen in QB64:

ScreenHandle& = _NewImage(800,600,32)

Screen ScreenHandle&

Here is the GDK method to create the same screen:

GDK_Screen_SetRes ScreenHandle, 800, 600, 32, 0

Once you have your screen setup you have successfully created the base of you program. It should look something like this:

GDK_Start

GDK_Screen_SetRes MainScreen, 640, 480, 32, 0

REM $INCLUDE:’UnseenGDK.bm’

User Input pt.1

UnseenGDK has built in support for logging both mouse and keyboard input. It does also support gamepads but only via use of the GDK_X library which is not covered in this tutorial.

Logging keyboard input.

In order to log keyboard input you must first create an array to hold the state of the keyboard keys. UnseenGDK has a prebuilt type array for this, KeyboardState.

To create a KeyboardState array (here named KB) do this:

DIM KB as KeyboardState

When you want to log the state of the keyboard keys do this: 

GDK_Keyboard_GetState KB

Now you have the state of most of the keyboard keys (not all of the keyboard keys are used in GDK, for a full list of available keys please see the TYPE array reference.) 

To check the state of the Left arrow key do this:

IF KB.Left THEN

‘// Your code here

END IF

If a key is down it returns  true (-1) or if it is released, false (0).

With UnseenGDK you can check the state of more than one key at a time. To check for the Left and Up arrow keys being pressed at the same time use this code:

IF KB.Left AND KB.Up THEN

‘// Your code here

END IF

Logging Mouse Input
To log mouse input, you use an almost identical method to logging the keyboard state. As with the keyboard, UnseenGDK has a predefined type array for the mouse, MouseState.

To create a MouseState array (here named MyMouse) do this:

DIM MyMouse as MouseState

When you want to log the state of the mouse buttons and mouse position do this: 

GDK_Mouse_GetState MyMouse

Now you have the state of the mouse stored in the MouseState array MyMouse.

To check the state of the Left mouse button do this:

IF MyMouse.LMB THEN

  ‘// Your code here

END IF

If a button is down it returns  true (-1) or if it is released, false (0). 

*(UnseenGDK only has support for the Left and Right mouse buttons.)

To print the position of the mouse do this:

Print MyMouse.MX

Print MyMouse.MY

So, you might be thinking, what’s wrong with the way I do it all ready? I doubt there is anything wrong with it, in fact I expect it’s an almost identical method, GDK just makes it easier to do.

Here’s how I would usually get the mouse state in QB64:

DIM Mx AS INTEGER, My AS INTEGER, LMB AS INTEGER, RMB AS INTEGER 

DO 

  MX = _MOUSEX

  MY = _MOUSEY

  LMB = _MOUSEBUTTON(1)

  RMB = _ MOUSEBUTTON(2)

LOOP WHILE _MOUSEINPUT

Compared to the GDK version it looks a bit excessive.

DIM MyMouse as MouseState

GDK_Mouse_GetState MyMouse

Drawing Sprites pt.1

There are several ways of drawing sprites in UnseenGDK, this is the most simple method.

In order to draw a sprite, you must first create an array to hold the sprites information and then load the sprites image. UnseenGDK has a predefined type for this, Sprite.

To create a sprite (here named Logo) do this:

DIM Logo AS Sprite
Now you need to load the sprites image and set it’s most basic information. 

GDK_Sprite_New Logo, “QBLogo.png”, 1, 1 , 1, 1

This loads an image called “QbLogo.png” and sets it basic parameters. This is what the GDK_Sprite_New command definition looks like in the library.

GDK_Sprite_New (SpriteFile AS Sprite, FileSource$, XFrameCount%, YFrameCount%, TotalFrameCount%, Scale!)
Now that you have loaded your sprite, before you can draw it you have to make it visible. This is done with the command GDK_Sprite_Show.

GDK_Sprite_Show Logo

Now all that’s left is to draw it. 

GDK_Sprite_DrawXY Logo, 100, 100, 0, 0

This would draw the image at x:100, y:100, with no rotation. This command can also handle animated sprites, but as we want to draw the entire image we use a 0.  Here is what the GDK_Sprite_DrawXY command looks like in the library:

GDK_Sprite_DrawXY (Sprite AS Sprite, X%, Y%, Rotation!, AnimationFrame%)
So this is what your code should look like:

GDK_Start

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0

DIM Logo AS SPRITE

GDK_Sprite_New Logo, “QBLogo.png”, 1, 1 , 1, 1

GDK_Sprite_Show Logo

GDK_Sprite_DrawXY Logo, 100, 100, 0, 0

REM $INCLUDE:’UnseenGDK.bm’

And that’s it, you can now draw sprites on the screen. Try experimenting with it to learn more about the command.

Vectors pt.1
UnseenGDK uses vectors to move objects automatically based on their direction (rotation angle) and speed. The rotation angle used is always in radians.

Although UnseenGDK supports rotation, it is not recommended that you draw no more than 12 rotated sprites at any one time otherwise your game speed may slow to a crawl. A new method is being designed to allow for almost unlimited rotated sprites but that will only be available via the GDK_X library and will not be covered here.

In order to use a vector, you must first create it and then set its variables. UnseenGDK contains a predefined type for this, Vector.

To create a vector array (here called BoxPos) do this:

DIM BoxPos AS Vector

Now that the Vector array has been created you need to set it’s initial variables:

GDK_Vector_New BoxPos, 100, 100, 0, 0

This creates a vector starting at x:100 y:100 with a speed of 0 and a rotation of 0. This would mean that unless you change it’s variables the vector will remain stationary.

Here is what the GDK_Vector_New command definition looks like: 

GDK_Vector_New (Vector AS Vector, X%, Y%, Speed!, Rotation#)
To see this working, try drawing a box at the vectors co-ordinates.

LINE (BoxPos.X, BoxPos.Y)-(BoxPos.X + 10, BoxPos.Y + 10), _RGB(200,200,200),BF

To make the vector move you need to apply force (speed) to the vector. This is done either by directly accessing the vectors variables, or by using the command GDK_Vector_SetSpeed.

GDK_Vector_SetSpeed BoxPos, 1

Now that the vector has force (speed), in order to make it move you need to call the command GDK_Vector_Move.

GDK_Vector_Move BoxPos

This would move the vector, but only once so calling GDK_Vector_Move each loop is advised.

Now for something a bit flashier, making the box move in a circle, here’s the complete code sample. 

GDK_Start '//Initialise GDK

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0 '// Create the program screen

DIM BoxPos AS Vector '// Create the Vector in memory

GDK_Vector_New BoxPos, 400, 300, 6, 7 * ATN(1) '// Set the vector's initial variables

DIM KB AS KeyBoardState

DO

  _LIMIT 30

  GDK_Keyboard_GetState KB 
  BoxPos.Rotation = BoxPos.Rotation + .1 '// Increase the rotation of the vector by .1 

  GDK_Vector_Move BoxPos '// Update the vectors position

  '//Draw a box at the Vectors position

  LINE (BoxPos.X, BoxPos.Y)-(BoxPos.X + 10, BoxPos.Y + 10), _RGB(0, 0, 200), BF

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC

SYSTEM '// If ESC was pressed exit the program

'// Include the GDK library.

REM $INCLUDE:'UnseenGDK.bm'

And that’s it for part one of vectors, more later.

Collision Detection pt.1

UnseenGDK has two methods of collision detection, Rectangle and BoundingCircle, this tutorial covers rectangle based collision detection.

To check for collisions between two sprites you must first create rectangles associated with the sprites position, size and orientation. You can create your own rectangles or have GDK create them for you. 

To create a Rectangle you must first create an array for it’s variables. 

DIM Rect1 as Rectangle

But one rectangle is of little use, we need two to check against each other.

DIM Rect2 as Rectangle

Now we have the two rectangles DIMensioned in memory, we need to set their variables.

To create the new rectangles do this:

GDK_Rectangle_New Rect1, 100, 100, 10, 10

GDK_Rectangle_New Rect2, 105, 105, 10, 10

This creates one rectangle at x:100, y:100 with a width of 10 pixels, and a height of 10 pixels, and another rectangle at x:105, y:105 with the same dimensions . To see if the two rectangles intersect, use the FUNCTION GDK_Rectangle_Intersect.
IF GDK_Rectangle_Intersect(Rect1, Rect2) THEN

  ‘// A collision 

ELSE
  ‘//Not a collision 
END IF

GDK_Rectangle_Intersect returns  true (-1)  if an intersection is happening, false (0) if not.

GDK Math

The math in UnseenGDK is based on the standard Cartesian layout. This is explained below.

Note :  4 * ATN(1) = Pi


[image: image1]
All rotation and vector movement in UnseenGDK is done based on this layout. As such all images that you wish to rotate and/or have move on vector lines should be normally orientated. Rotation is always measured in radians.
[image: image2.png]Z‘A;E



 This is the correct orientation.

 This is an incorrect orientation.

True and False
UnseenGDK uses the same method of determining if a value is true or false as QB64. 

True = -1, False = 0
So if your checking a variable to be true you can do this:

IF AVarible% THEN

‘// Code to run if true

END IF

And if you want to check if a variable is false then you can do this:

IF NOT AVarible% THEN

‘// Code to run if false

END IF
GDK_MathX

Calling this function allows you to use the C++ math library functions listed below.

atan2# (BYVAL y#, BYVAL x#)

acos# (BYVAL x#)

asin# (BYVAL x#)

cosh# (BYVAL x#)

sinh# (BYVAL x#)

tanh# (BYVAL x#)

pow# (BYVAL base#, BYVAL exponent#)

Game 1 – Bounce
This program will simulate a bouncing ball, the user will be able to control a paddle for the ball to bounce against at the  bottom of the screen. The idea of the game is never let the ball drop of the screen. 

So, before you even think about writing a program/game. You should think about all of the elements in the program/game, and how they move and interact with one another. Trying to get it all working at once will be near impossible. It is recommended you add one element at a time. There are only two elements in this game, a ball and a paddle.

First you need to set up your program. Here is what an UnseenGDK program should look like before you add any elements or controls:

GDK_Start '//Initialise GDK

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0 '// Create the program screen

DIM KB AS KeyBoardState '// Create a KeyboardState array for checking key presses

DO ‘ // Main game loop

  _LIMIT 60

  GDK_Keyboard_GetState KB '//Get the state of the keyboard.

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC ‘// End of main game loop

SYSTEM '// If ESC was pressed exit the program

'// Include the GDK library.

REM $INCLUDE:'UnseenGDK.bm'

You might want to save this code so you don’t have to write it every time you start a new GDK project.
Adding the user paddle

So the first element you should make is the paddle. This will be drawn in this tutorial with the built in QB64 drawing commands, though you could use a sprite instead.

The paddle will need to move and also will need to be checked for collisions against the ball, this means it will need both a Vector array to hold it’s position and a Rectangle array to use for checking collisions.

DIM PaddlePos AS Vector, PaddleRect AS Rectangle

At the start of the game, the paddles top left corner will be at x: 360, y: 560, it will be 80 pixels wide and 30 pixels high, the paddle will be full blue RGB(0,0,255). 

To set the paddles initial position, use the GDK_Vector_New command:

GDK_Vector_New PaddlePos, 360, 560, 0, 0

Now you can use the vector array to draw the paddle:

LINE (PaddlePos.X, PaddlePos.Y)-(PaddlePos.X + 80, PaddlePos.Y +30),  _RGB(0,0,255), BF

In this game we will use the left and right arrow keys to move the paddle. 

You need to remember to check that the paddle is still on the screen before you move it!

To use the keyboard to move the paddle:

GDK_Keyboard_GetState KB

IF KB.Left THEN

  IF PaddlePos.X > 0 THEN PaddlePos.X = PaddlePos.X - 2

ELSEIF KB.Right THEN

  IF PaddlePos.X + 80 < 800 THEN PaddlePos.X = PaddlePos.X + 2

END IF

And finally, making the paddles collision rectangle, this bits a lot easier, but make sure you include a call to this function each loop to ensure the rectangle is at the same position as the paddle.

To create the collision rectangle at the same position and size as the paddle:

GDK_Rectangle_New PaddleRect, PaddlePos.X, PaddlePos.Y, 80, 30

We don’t have another rectangle to check the paddle against just yet, so we will add that later. And that’s it, the paddle has been added and you can move it with the arrow keys, your program should look something like this.

GDK_Start '//Initialise GDK

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0 '// Create the program screen

DIM KB AS KeyBoardState '// Create a KeyboardState array for checking key presses

DIM PaddlePos AS VECTOR, PaddleRect AS RECTANGLE

GDK_Vector_New PaddlePos, 360, 560, 0, 0

DO ‘ // Main game loop

  _LIMIT 60

  GDK_Keyboard_GetState KB '//Get the state of the keyboard.

  IF KB.Left THEN

    IF PaddlePos.X > 0 THEN PaddlePos.X = PaddlePos.X - 2

  ELSEIF KB.Right THEN

    IF PaddlePos.X + 80 < 800 THEN PaddlePos.X = PaddlePos.X + 2

  END IF

  GDK_Rectangle_New PaddleRect, PaddlePos.X, PaddlePos.Y, 80, 30

  LINE (PaddlePos.X, PaddlePos.Y)-(PaddlePos.X + 80, PaddlePos.Y +30),  _RGB(0,0,255), BF

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC ‘// End of main game loop

SYSTEM '// If ESC was pressed exit the program

'// Include the GDK library.

REM $INCLUDE:'UnseenGDK.bm'

I bet your thinking “Wow, that’s so cool!” ( Well, ok, it’s not that impressive, but then again there’s not a lot of code.

Adding the ball

Let’s add the ball to the program. As with the paddle the ball will need a Vector array and a Rectangle array. 

DIM BallPos AS VECTOR, BallRect AS RECTANGLE

At the start of the game the ball will be in the middle of the screen and always move at a speed of 4. Its initial direction will be up and right (45 degrees). 

GDK_Vector_New BallPos, 400, 300, 4, 7 * ATN(1)

Now we have a co-ordinate for the ball, lets draw it.

CIRCLE (BallPos.X, BallPos.Y), 8, _RGB(255,255,255)

PAINT STEP(0, 0), _RGB(255,255,255), _RGB(255,255,255)

This will draw a filled white circle with a radius of 8 pixels.

Now you need to create a collision rectangle for the ball based on its size and position.

GDK_Rectangle_New BallRect, BallPos.X – 8, BallPos.Y – 8, 16, 16

As the ball needs to move on it’s own it needs to update its position each loop, this is a little more complicated than moving the paddle as we need to account for 4 different boundaries, and to also check for collisions with the paddle.

If you wanted to just move the ball:

GDK_Vector_Move BallPos

However this would ignore the limits of the screen and is of little use without first checking the balls position.

The two most important checks are: Between the ball and the paddle, and between the ball and the bottom of the screen so you should do these first. 


To check the balls position against the paddles, the easiest way is to use the rectangle intersection method. If there is no collision, the program should then do a boundary check.

Here is the complete code for checking the balls position and reacting to it.

    IF GDK_Rectangle_Intersect(BallRect, PaddleRect) THEN

    '// A collision occurred, change the balls direction.

    IF BallPos.Rotation = 3 * ATN(1) THEN

      BallPos.Rotation = 1 * ATN(1)

    ELSEIF BallPos.Rotation = 5 * ATN(1) THEN

      BallPos.Rotation = 7 * ATN(1)

    END IF

  ELSE

    '// No collision - do boundary checks

    IF BallPos.Y < 600 THEN '// The ball is above the bottom of the screen.

      IF BallPos.X - 8 > 0 THEN ‘// The Ball is not at the left edge

        IF BallPos.X + 8 < 800 THEN ‘// the ball is not at the right edge

          IF BallPos.Y - 8 <= 0 THEN ‘// The ball is above or equal to the top edge

            IF BallPos.Rotation = 1 * ATN(1) THEN

              BallPos.Rotation = 3 * ATN(1)

            ELSEIF BallPos.Rotation = 7 * ATN(1) THEN

              BallPos.Rotation = 5 * ATN(1)

            END IF

          END IF

        ELSE

          IF BallPos.Rotation = 7 * ATN(1) THEN

            BallPos.Rotation = 1 * ATN(1)

          ELSEIF BallPos.Rotation = 5 * ATN(1) THEN

            BallPos.Rotation = 3 * ATN(1)

          END IF

        END IF

      ELSE

        IF BallPos.Rotation = 1 * ATN(1) THEN

          BallPos.Rotation = 7 * ATN(1)

        ELSEIF BallPos.Rotation = 3 * ATN(1) THEN

          BallPos.Rotation = 5 * ATN(1)

        END IF

      END IF

    ELSE

      '// Reset the ball and the paddle 

      BallPos.X = 400

      BallPos.Y = 300

      BallPos.Rotation = 7 * ATN(1)

      PaddlePos.X = 360

    END IF

  END IF

  GDK_Vector_Move BallPos

It might seem like a lot just to make a ball bounce around, but making games is a long process and you should be prepared to have to write a lot of code to make your game complete.

Now you have a completed game, here is what it should look like:

*** You can copy/paste this code into the QB64 IDE and press F5 to try it.

GDK_Start '//Initialise GDK

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0 '// Create the program screen

DIM KB AS KeyBoardState '// Create a KeyboardState array for checking key presses

DIM PaddlePos AS Vector, PaddleRect AS Rectangle

DIM BallPos AS Vector, BallRect AS Rectangle

GDK_Vector_New PaddlePos, 360, 560, 0, 0

GDK_Vector_New BallPos, 400, 300, 4, 7 * ATN(1)

DO ' // Main game loop

  _LIMIT 60

  GDK_Keyboard_GetState KB '//Get the state of the keyboard.

  IF KB.Left THEN

    IF PaddlePos.X > 0 THEN PaddlePos.X = PaddlePos.X - 2

  ELSEIF KB.Right THEN

    IF PaddlePos.X + 80 < 800 THEN PaddlePos.X = PaddlePos.X + 2

  END IF

  GDK_Rectangle_New PaddleRect, PaddlePos.X, PaddlePos.Y, 80, 30

  LINE (PaddlePos.X, PaddlePos.Y)-(PaddlePos.X + 80, PaddlePos.Y + 30), _RGB(0, 0, 255), BF

  IF GDK_Rectangle_Intersect(BallRect, PaddleRect) THEN

    '// A collision occurred, change the balls direction.

    IF BallPos.Rotation = 3 * ATN(1) THEN

      BallPos.Rotation = 1 * ATN(1)

    ELSEIF BallPos.Rotation = 5 * ATN(1) THEN

      BallPos.Rotation = 7 * ATN(1)

    END IF

  ELSE

    '// No collision - do boundary checks

IF BallPos.Y < 600 THEN '// The ball is above the bottom of the screen.

      IF BallPos.X - 8 > 0 THEN ‘// The Ball is not at the left edge

        IF BallPos.X + 8 < 800 THEN ‘// the ball is not at the right edge

          IF BallPos.Y - 8 <= 0 THEN ‘// The ball is above or equal to the top edge

            IF BallPos.Rotation = 1 * ATN(1) THEN

              BallPos.Rotation = 3 * ATN(1)

            ELSEIF BallPos.Rotation = 7 * ATN(1) THEN

              BallPos.Rotation = 5 * ATN(1)

            END IF

          END IF

        ELSE

          IF BallPos.Rotation = 7 * ATN(1) THEN

            BallPos.Rotation = 1 * ATN(1)

          ELSEIF BallPos.Rotation = 5 * ATN(1) THEN

            BallPos.Rotation = 3 * ATN(1)

          END IF

        END IF

      ELSE

        IF BallPos.Rotation = 1 * ATN(1) THEN

          BallPos.Rotation = 7 * ATN(1)

        ELSEIF BallPos.Rotation = 3 * ATN(1) THEN

          BallPos.Rotation = 5 * ATN(1)

        END IF

      END IF

    ELSE

      '// Reset the ball and the paddles position

      BallPos.X = 400

      BallPos.Y = 300

      BallPos.Rotation = 7 * ATN(1)

      PaddlePos.X = 360

    END IF

  END IF

  GDK_Vector_Move BallPos

  CIRCLE (BallPos.X, BallPos.Y), 8, _RGB(255, 255, 255)

  PAINT STEP(0, 0), _RGB(255, 255, 255), _RGB(255, 255, 255)

  GDK_Rectangle_New BallRect, BallPos.X - 8, BallPos.Y - 8, 16, 16

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC '// End of main game loop

SYSTEM '// If ESC was pressed exit the program

'// Include the GDK library.

REM $INCLUDE:'UnseenGDK.bm'

You should try adding a score and lives counter to the game to improve your knowledge of how it works. Maybe even try increasing the balls speed every time the player scores a certain number of points to add an extra element to the game play. 
This basic game can be used as the core of many other games such as Pong, Breakout, and, with a little expansion, Space invaders! (
Drawing sprites pt.2

As mentioned in drawing sprites pt.1, UnseenGDK has several commands that allow you to draw sprites. This tutorial covers drawing sprites based on a vector, using the GDK_Sprite_Draw command.

First initialise the program as normal, then dimension and create Sprite and Vector arrays for the image that your going to draw. 

Your code should look something like this:

GDK_Start

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0

DIM KB AS KeyboardState. LogoImg AS Sprite, LogoPos AS Vector

GDK_Vector_New LogoPos, 300, 400, 0, 0

GDK_Sprite_New LogoImg, “QbLogo.png”, 1,1,1,1

GDK_Sprite_Show LogoImg

DO ‘ // Main game loop

  _LIMIT 60

  GDK_Keyboard_GetState KB 

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC ‘// End of main game loop

SYSTEM 

REM $INCLUDE:’UnseenGDK.bm’

You could draw the sprite with the same method as in the first sprite drawing tutorial, using GDK_Sprite_DrawXY. There is no reason why you can’t do it with this method, however it is easier and faster to use the GDK_Sprite_Draw command. 

* Unlike GDK_Sprite_Draw,  GDK_Sprite_DrawXY can be used to draw a sprite without any rotation and still allow you to use vector movement for the associated sprite. This can really help with game speed in certain situations, but more on that later. 

Here is what the GDK_Sprite_Draw command looks like:

GDK_Sprite_Draw (SpriteFile AS Sprite, DestVector AS Vector, Frame%)
So to draw the image at the specified vector, do this:

GDK_Sprite_Draw LogoImg, LogoPos, 0

This will draw the image at x:300, y:400, with no rotation. 

That’s all well, and good, but I can hear you thinking “_Putimage can do that!”, and yes it can. In fact if you have an object with no rotation, GDK draws it on the screen with _putimage.
Say you want to make the image rotate by pressing either left or right arrow keys, increase speed when the up arrow is pressed and decrease speed when the down arrow is pressed. 
To do this you will need to add control code that responds to keyboard input and use the GDK_Vector_Move command to move the sprite based on its speed and rotation.
It may seem a bit backwards, but you need to remember this. To rotate left, increase the vectors rotation value, to rotate right, decrease it. 

So here’s what your code should look like:
*** You can copy/paste this code into the QB64 IDE and press F5 to try it.

GDK_Start

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0

DIM KB AS KeyBoardState, LogoImg AS Sprite, LogoPos AS Vector

GDK_Vector_New LogoPos, 300, 400, 0, 0

GDK_Sprite_New LogoImg, "QbLogo.png", 1, 1, 1, 1

GDK_Sprite_Show LogoImg

DO ' // Main game loop

  _LIMIT 60

  GDK_Keyboard_GetState KB

  IF KB.Left THEN

    LogoPos.Rotation = LogoPos.Rotation + .1

  ELSEIF KB.Right THEN

    LogoPos.Rotation = LogoPos.Rotation - .1

  END IF

  IF KB.Up THEN

    IF LogoPos.Speed < 6 THEN LogoPos.Speed = LogoPos.Speed + 1

  ELSEIF KB.Down THEN

    IF LogoPos.Speed > 0 THEN LogoPos.Speed = LogoPos.Speed - 1

  END IF

  GDK_Vector_Move LogoPos

  GDK_Sprite_Draw LogoImg, LogoPos, 0

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC '// End of main game loop

SYSTEM

REM $INCLUDE:'UnseenGDK.bm'

The sprite will move based on its assigned vectors speed and rotation. You may notice that it rotates differently when rotating left to when it rotates right. This is because the sprite is rotated from the top left corner of the image. You can change this though.

Add this line after the GDK_Sprite_Show command and re-run the program.

GDK_Sprite_SetRotationPoint LogoImg, LogoImg.Width / 2, LogoImg.Height / 2

This sets the sprites rotation point to the centre of the image.

Drawing animated sprites pt.1

Up until now we have only drawn images using the built in QB64 commands or using single sprites. UnseenGDK has the ability to use equally spaced sprite sheets automatically. Here’s a typical sprite sheet:

As you can see it shows the earth in different stages of rotation. If you were to break the image up into 12 separate images and display them one at a time, you could create a convincing effect. 

By now you should be comfortable in creating single frame sprites. The method to creating an animated sprite is identical, except you provide the number of animation frames in x and y directions. The sprite above contains 12 images in a 4 * 3 grid.

So here’s how we would create this sprite:

DIM Earth AS Sprite

GDK_Sprite_New Earth, “Earth.png”, 4, 3, 12, 1

GDK_Sprite_Show Earth

I will explain this a bit more:

GDK_Sprite_New Earth, “Earth.png”, 4, 3, 12, 1

This will load the image Earth.png into the Earth type array. It sets it to have 4 frames on the x axis and 3 frames on the y axis, with a total of 12 frames. The 1 on the end is the scale you wish to draw the image.

To draw the sprite you can use either the GDK_Sprite_Draw and GDK_Sprite_DrawXY commands, the choice is yours, I will demonstrate the GDK_Draw_SpriteXY method. This is because we don’t need the earth to move so creating a vector would be a waste of time and memory.

Now that the sprite sheet has been created you can draw any frame from the image you wish just by stating the number of the frame in the draw command.

Sprites are numbered as such.


[image: image5]
So lets make the earth animated, here’s the complete code sample:

*** You can copy/paste this code into the QB64 IDE and press F5 to try it.

GDK_Start

GDK_Screen_SetRes MainScreen, 800, 600, 32, 0

'// Create a type array to store the information on the sprite.

DIM Earth AS Sprite

'// Load the sprites image and set. it's initial parameters

GDK_Sprite_New Earth, "Earth.png", 4, 3, 12, 1

GDK_Sprite_Show Earth

DIM KB AS KeyBoardState

DO

  _LIMIT 60

  GDK_Keyboard_GetState KB '//Get the state of the keyboard.

  '// Increase or reset the animation each loop

  IF AnimFrame% < 12 THEN AnimFrame% = AnimFrame% + 1 ELSE AnimFrame% = 1

  '// Draw the sprite using the animation counter

  '// to decide which frame to display

  GDK_Sprite_DrawXY Earth, 0, 0, 0, AnimFrame%

  _DELAY 0.2

  _DISPLAY

  CLS

LOOP UNTIL KB.ESC

SYSTEM

REM $INCLUDE:'UnseenGDK.bm'

And that’s the basics of drawing animated sprites. We will look at these again later, for now try making the sprite rotate in reverse, be drawn at a different size or change frames at a different speed to learn how the code works.

Tile engine pt.1
Games like Commander Keen, Mario, Sonic, Zelda and Final Fantasy can all be made with the use of a tile engine. Each tile used in the game to make the level is amalgamated into one image. Here is a basic tile sheet for platform games:

As you can see the tiles are universally sized and spaced. 
Using the tile engine is a little more complicated than anything I have covered so far, but learning how to use it can really help in both designing levels and the speed of your tile based games.
In UnseenGDK each tile is referenced by number.

[image: image7]
Before you can use the tile engine you need to initialise it’s type arrays. This command would be placed after the call to GDK_Start. 
The start of your program, should look like this:

GDK_Start

GDK_TileEngine_Init

Now you can access the types and use the functions in the tile engine.

There are 2 elements to the tile engine, the Tile set and the Tile map. We will look at the tile set first. 

The tile set incorporates two type arrays. Tileset and XY. Tileset is used to store the information on the actual image, whereas an array of XY is used to store each tiles position in the image. 
When you load the tile sheet each tiles position is calculated and stored in the XY array, this array must be the same size as the total number of possible tiles in your tile sheet. This makes it a lot faster than calculating each tiles position every time you want to draw one.
The tile sheet shown has 85 possible tiles in a 17 * 5 grid. So in order to store the X/Y position of each tile you would need an array with at least 85 possible records. 
XY arrays are ALWAYS numbered from 1.

Loading a tile sheet

I will be using the tile sheet shown on the previous page for this tutorial. 

So to load a tile sheet you must first DIMension a Tileset array and an array of the XY type.
DIM TileSheet1 AS Tileset, TileSheet1XY(1 TO 85) AS XY

Type array reference
These types are initiated when you call GDK_Start


TYPE Sprite

  File AS LONG

  Width AS INTEGER

  Height AS INTEGER

  Alpha AS LONG

  IsVisible AS INTEGER

  XFrameCount AS INTEGER

  YFrameCount AS INTEGER

  TotalFrameCount AS INTEGER

  RotationX AS INTEGER

  RotationY AS INTEGER

  Scale AS SINGLE

TYPE Vector

  X AS SINGLE

  Y AS SINGLE

  Speed AS SINGLE

  Rotation AS DOUBLE

  Accel AS SINGLE

  Decel AS SINGLE

TYPE Rectangle

  X AS INTEGER

  Y AS INTEGER

  Width AS INTEGER

  Height AS INTEGER

  CollisionX AS INTEGER

  CollisionY AS INTEGER

  CollisionXX AS INTEGER

  CollisionYY AS INTEGER

TYPE BoundingCircle

  X AS INTEGER

  Y AS INTEGER

  Radius AS INTEGER

TYPE MouseState

  Mx AS INTEGER

  My AS INTEGER

  LMB AS INTEGER

  RMB AS INTEGER

TYPE GameObject

  Sprite AS Sprite

  Vector AS Vector

  Rect AS Rectangle
Command reference

Initialisation commands

GDK_Start


Initialises the GDK base types.

GDK_Particles_Init


Initialise the types associated with the particle engine.

GDK_TileEngine_Init


Initialises the types associated with the tile engine.

Collision detection commands

GDK_Rectangle_New (RectRef AS Rectangle, x%, y%, Width%, Height%)


Sets a new collision rectangles position and size.

GDK_Rectangle_Intersect (RectangleRef1 AS Rectangle, RectangleRef2 AS Rectangle)


Returns -1 if two rectangles intersect, 0 if they don’t.

GDK_GetCollisionArea (RectRef1 AS Rectangle, RectRef2 AS Rectangle)

Calculates and stores the exact collision area between two rectangles. This can be used to make pixel perfect collision detection functions.

GDK_Rectangle_AutoSize (TextRef AS Sprite, Vector AS Vector, Rect AS Rectangle)

Creates a collision rectangle based on a sprite and vector, this command does not support animated sprites.

GDK_BoundingCircle_New (Circ1 AS BoundingCircle, X%, Y%, Radius%)


Sets a new BoundingCircles position and size.

GDK_BoundingCircle_Intersect (Circ1 AS BoundingCircle, Circ2 AS BoundingCircle)

Returns -1 if two circles intersect, 0 if they don’t.

Vector commands

GDK_Vector_New (Vector AS Vector, X%, Y%, Speed!, Rotation#)


Sets a vectors initial values.

GDK_Vector_SetPosition (Vector AS Vector, X%, Y%)


Sets the X/Y position of the vector.

GDK_Vector_SetRotation (Vector AS Vector, Rotation#)


Sets the rotation of the vector in radians. 

GDK_Vector_SetSpeed (Vector AS Vector, Speed!)


Sets the speed of a vector.

GDK_Vector_Update (Vector AS Vector)

Moves a vector based on its direction (rotation) and force (speed), but takes acceleration and deceleration into account before calculating the new position and speed. Deceleration will have no effect once the speed and acceleration values are 0.
GDK_Vector_SetAccelDecel (Vector AS Vector, Accel!, Decel!)

Sets the acceleration and deceleration values of a vector. When you call GDK_Vector_Update, the deceleration value is subtracted from the acceleration value until the acceleration value is 0, then the deceleration value is subtracted from the speed value until it reaches 0.
GDK_Vector_Move (Vector AS Vector)


Moves a vector based on its direction (rotation) and force (speed).

Sprite commands

GDK_Sprite_New (SpriteFile AS Sprite, FileSource$, XFrameCount%, YFrameCount%, TotalFrameCount%, Scale!)

Loads the image for the sprite and sets it’s primary values.

GDK_Sprite_SetRotationPoint (SpriteFile AS Sprite, X%, Y%)


Sets the rotation point of a sprite.

GDK_Sprite_SetVisibility (SpriteFile AS Sprite, OnOff%)

Sets a sprites visibility, -1 for on, 0 for off.

GDK_Sprite_SetAlpha (SpriteFile AS Sprite, Alpha&)


Sets the alpha value (transparent colour) of the specified sprite.

GDK_Sprite_DrawXY (Sprite AS Sprite, X%, Y%, Rotation!, AnimationFrame%)


Draws a sprite at a specified position, with options for rotation and animation.

GDK_Sprite_Draw (SpriteFile AS Sprite, DestVector AS Vector, Frame%)


Draws a sprite at a specified vector, with an option for animation.

GDK_Sprite_Hide (Handle AS Sprite)
Makes a specified sprite invisible.

GDK_Sprite_Show (Handle AS Sprite)
Makes a specified sprite visible.

GDK_Sprite_Free (SpriteFile AS Sprite)


Frees a specified sprites image.
GDK_Sprite_Rotate (TypeRef AS Sprite, sx!, sy!, Vector AS Vector, Scale!)

Rotates a specified sprite. This command is not need except for making you own custom drawing commands (Very advanced GDK).
Screen commands

*UnseenGDK always uses 32 Bit mode when using any drawing commands, do not try and load an image, draw particles or attempt any other graphics function in any other screen mode.

GDK_Screen_MatchRes (ScreenHandle&, clr%, fs%)

Matches the current screen size, with options for colour mode and full screen mode.

Example :

GDK_Screen_MatchRes MainScreen, 32, 0

GDK_Screen_SetRes (ScreenHandle&, width%, height%, clr%, fs%)

GDK_WindowHeight%

GDK_WindowWidth%

GDK_ScreenWidth%

GDK_ScreenHeight%

GDK_ToggleFullScreen

GDK_Screen_Store (ImageHandle&)

GDK_Screen_Restore (ImageHandle&)

GDK_Screen_FadeIn (Time#)

GDK_Screen_FadeOut (Time#)

Input commands

GDK_Keyboard_GetState (KeyboardRef AS KeyBoardState)


Stores the state of the keyboard keys in the KeyboardState array.

GDK_Mouse_GetState (MouseRef AS MouseState)


Stores the state of the mouse in the MouseState array.

Graphics commands

GDK_ApplyMonoFilter (TypeRef AS Sprite)

Makes a specified sprite into a monochrome image.

GDK_ApplyColorFilter (TypeRef AS Sprite, r%, g%, b%, a%)
GDK_ReplaceColor (TypeRef AS Sprite, OldColor&, NewColor&)


Replaces every pixel the matches OldColor& with NewColor&.

GDK_ChangeColorValue (TypeRef AS Sprite, Color$, Value%)

GDK_Sphere (X%, Y%, Radius%, R%, G%, B%)


Draws an alpha blended sphere at the specified location.
Particle engine commands

GDK_Emitter_New (EmitRef AS Emitter, x%, y%, SpawnRate%, AngleMin!, AngleMax!, VelMin!, VelMax!, MinLifeTime#, MaxLifeTime#)

GDK_Emitter_SetAccelDecel (EmitRef AS Emitter, AccelMin!, AccelMax!, DecelMin!, DecelMax!)

GDK_Particles_Spawn (EmitRef AS Emitter, PartRef() AS Particle, ParticleMax%, InitialColor&, ColorChange&, ChangeMode%)

GDK_Particles_Update (PartRef() AS Particle, ParticleMax%)

GDK_Particles_Draw (PartRef() AS Particle, ParticleMax%)

Math commands

GDK_Distance! (X%, Y%, XX%, YY%)


Returns the distance in pixels between two points.

GDK_RadianToDegree! (Radians!)


Returns the angle in degrees of an angle in radians.
GDK_DegreeToRadian! (Degrees!)


Returns the angle in radians of an angle in degrees.

GDK_MathX


Allows access to the C++ math library.
Tile engine commands

SUB GDK_TileMap_Load (Level() AS Tile, XTiles%, YTiles%)

SUB GDK_Tiles_Draw (TileRef AS Tileset, XYRef() AS XY, Level() AS Tile, X%, Y%, XCnt%, YCnt%, XMax%, YMax%)

SUB GDK_Tileset_New (TileRef AS Tileset, File$, XYRef() AS XY, XTiles%, YTiles%, Alpha&, XOffset%, YOffset%)

SUB GDK_Tile_Draw (TileRef AS Tileset, TileNum%, XYRef() AS XY, x%, y%)

Game object commands

GDK_GameObject_New (Handle AS GameObject, File$, XframeCount%, YFrameCount%, X%, Y%,Speed!, Rotation!)

GDK_GameObject_Draw (GameObject AS GameObject, AnimationFrame%)

GDK_GameObject_Update (Handle AS GameObject)

GDK_GameObject_Rotate (Handle AS GameObject, Rot!)

GDK_GameObject_Hide (Handle AS GameObject)

GDK_GameObject_Show (Handle AS GameObject)

GDK_GameObject_Intersect (Handle AS GameObject, Handle2 AS GameObject)

What is GDK_X?
GDK_X is a library which relies on UnseenGDK. It offers a whole range of commands that are either not directly used in making games,  rely upon external c++ libraries or are experimental commands that are not yet complete. 
Unlike UnseenGDK, GDK_X is a changing library that will have updates.
What it has now : 


Gamepad support (via SDL)


Various graphics routines


Image drawing and manipulation commands


Blitting functions


Sound commands


Other SDL bindings for custom mouse routines and screen manipulation.

Is there anything else with UnseenGDK?

Yes, there is GDK_Draw , a library that contains shape drawing commands. It does not rely on UnseenGDK, and one day there will be UnseenGDK3D, but that will be a  long time coming. 
Here’s a list of the current commands in GDK_Draw: 
GDK_Triangle (x1%, y1%, x2%, y2%, x3%, y3%, CLR&)

GDK_TriangleFill (x1%, y1%, x2%, y2%, x3%, y3%, CLR&, FillCLR&)

GDK_Ellipse (X%, Y%, XRadius!, YRadius!, CLR&)

GDK_EllipseFill (X%, Y%, XRadius!, YRadius!, CLR&, FillCLR&)

GDK_EllipseSegment (X%, Y%, XRadius!, YRadius!, Start!, End!, CLR&)

GDK_EllipseSegmentFill (X%, Y%, XRadius!, YRadius!, Start!, End!, CLR&, FillClr&)

GDK_Circle (X%, Y%, Radius!, CLR&)

GDK_CircleFill (X%, Y%, Radius!, CLR&, FillCLR&)

GDK_CircleSegment (X%, Y%, Radius!, CLR&, CircleStart!, CircleEnd!)

GDK_CircleSegmentFill (X%, Y%, Radius!, CLR&, FillClr&, CircleStart!, CircleEnd!)

GDK_Line (X%, Y%, XX%, YY%, CLR&, Wide%)

GDK_LineAngle (X%, Y%, Angle!, Length!, CLR&)

GDK_LineH (X%, XX%, Y%, CLR&, Wide%)

GDK_LineV (Y%, YY%, X%, CLR&, Wide%)

GDK_Box (X%, Y%, Wide%, High%, CLR&)

GDK_BoxFill (X%, Y%, Wide%, High%, CLR&, FillCLR&)

GDK_BoxXS (X%, Y%, Wide%, High%, XWide%, CLR&, ShadeClr&)

GDK_EllipseXS (X%, Y%, XRadius!, YRadius!, CLR&, Wide%, ShadeClr&)

GDK_CircleXS (X%, Y%, Radius!, CLR&, Wide%, ShadeClr&)

GDK_EllipseX (X%, Y%, XRadius!, YRadius!, CLR&, Wide%)

GDK_CircleX (X%, Y%, Radius!, CLR&, Wide%)

TYPE KeyBoardState


  Left AS LONG


  Right AS LONG


  Down AS LONG


  Up AS LONG


  CTRL AS LONG


  SHIFT AS LONG


  ALT AS LONG


  SPACE AS LONG


  ENTER AS LONG


  ESC AS LONG


  Num1 AS LONG


  Num2 AS LONG


  Num3 AS LONG


  Num4 AS LONG


  Num5 AS LONG


  Num6 AS LONG


  Num7 AS LONG


  Num8 AS LONG


  Num9 AS LONG


  Num0 AS LONG


  PLUS AS LONG


  MINUS AS LONG


  BACKSPACE AS LONG


  TAB AS LONG


  A AS LONG


  B AS LONG


  C AS LONG


  D AS LONG


  E AS LONG


  F AS LONG


  G AS LONG


  H AS LONG


  I AS LONG


  J AS LONG


  K AS LONG


  L AS LONG


  M AS LONG


  N AS LONG


  O AS LONG


  P AS LONG


  Q AS LONG


  R AS LONG


  S AS LONG


  T AS LONG


  U AS LONG


  V AS LONG


  W AS LONG


  X AS LONG


  Y AS LONG


  Z AS LONG

